Lab 10: Taylor Polynomials, Series, and Remainder

Taylor Polynomials

A Taylor polynomial is an approximation of a function. The n^{th} degree Taylor polynomial is given by

$$\begin{aligned} T_n(x) &= \sum_{k=0}^{n} \frac{f^{(k)}(a)}{k!}(x-a)^k \\ &= \frac{f(a)}{0!}(x-a)^0 + \frac{f'(a)}{1!}(x-a)^1 + \frac{f''(a)}{2!}(x-a)^2 + \cdots + \frac{f^{(n)}}{n!}(x-a)^n \end{aligned}$$

where $f^{(k)}$ denotes the k^{th} derivative of f. Also, it is the Taylor polynomial about $x = a$ (or "centered" at $x = a$). If we let n go to infinity, the full Taylor Series is:

$$\sum_{k=0}^{\infty} \frac{f^{(k)}(a)}{k!}(x-a)^k$$

Mathematica Calculates Taylor Polynomials

As in the previous lab, we can use the Mathematica command `Series` to get the Taylor polynomial for a function. Here, the syntax is `Series[f[x], {x, a, b}]` if we want to know the Taylor polynomial for a function `f[x]` at $x = a$, up to the term x^b (i.e., the b-th degree Taylor polynomial). So, for instance, consider the following code:

```
f[x_]:= 1/(Sqrt[x^4 + x^3 + 1])
T3[x_] = Series[f[x], {x, 0, 3}] // Normal
Plot[{f[x], T3[x]}, {x, -1.5, 1.5}, PlotLegends -> "Expressions"]
```

We get that the function `T3` is equal to $1-x^3/2$, which is the third degree Taylor polynomial for our function $1/\sqrt{x^4+x^3+1}$. The command `// Normal` was added to get rid of the Big O term at the end of the series approximation. If you look at the plot, you can see how the two curves are virtually identical very close to $x = 0$. Let's try this again, but this time, we'll find the Taylor polynomial at $x = 1$.

```
Clear[f,x]
f[x_]:= 1/(Sqrt[x^4 + x^3 + 1])
T3v2[x_] = Series[f[x], {x, 1, 3}] // Normal
Plot[{f[x], T3v2[x]}, {x, -1, 2}, PlotLegends -> "Expressions"]
```

`T3v2[x]` is a different function from `T3[x]`, and the curve for `T3v2[x]` is closer to the curve for `f[x]` near $x = 1$ instead of near $x = 0$.
Let's find some more Taylor polynomials to approximate this function, and plot them. This time, we'll calculate and plot Taylor polynomials at $x = -1$.

```
Clear[f, x, T3, T3v2]
f[x_] := 1/(Sqrt[x^4 + x^3 + 1])
T3[x_] = Series[f[x], {x, -1, 3}] // Normal
T5[x_] = Series[f[x], {x, -1, 5}] // Normal
T7[x_] = Series[f[x], {x, -1, 7}] // Normal
T9[x_] = Series[f[x], {x, -1, 9}] // Normal
Plot[{f[x], T3[x], T5[x], T7[x], T9[x]}, {x, -2, 2},
PlotLegends -> "Expressions", PlotRange -> {-1, 2}]
```

Mathematica guesses general term formula

The Series function can only expand the function into first several terms in Taylor series. Yet in Mathematica, to determine the series' convergence needs a general formula, how do we come by that? Mathematica provides a function called `FindSequenceFunction` which will guess a general term formula given a small number of terms in the sequence. Before doing that, we need to procure enough number of Taylor coefficients for Mathematica to work with. The combination of `Table` and `SeriesCoefficient` generate a list of Taylor coefficients.

```
Clear[f, x]
f[x_] := x Exp[-2x]
TaylorList = Table[SeriesCoefficient[f[x], {x, 0, n}], {n, 0, 10}]
```

Then we feed the list into function `FindSequenceFunction` to guess.

```
a[n_] = FindSequenceFunction[TaylorList, n]
```

Now `a[n_]` is the general formula for n-th term. We apply the `SumConvergence` function to find the convergence interval of the power series $\sum_{n=1}^{\infty} a[n]x^n$.

```
SumConvergence[a[n]x^n,n]
```

Notice that if the expansion center is not 0, you will have to change the x^n part.

The Remainder Term

As you can see, Taylor polynomials give us an approximation, but these approximations are rarely exact. There's usually something left over. That left over stuff is called the remainder $R_n(x)$, which is defined more precisely as

$$R_n(x) = f(x) - T_n(x)$$

where $T_n(x)$ is the n^{th}-degree Taylor polynomial for f. According to Taylor's theorem, this remainder satisfies

$$R_n(x) = \frac{f^{(n+1)}(c)}{(n+1)!}(x-a)^{n+1}$$

where $f^{(n+1)}$ is the $(n+1)^{\text{th}}$ derivative of f, and c is some number between x and a.

Remainder Example

Let's try to approximate sin(0.3) by calculating the value of the third degree Taylor polynomial about $x = 0$ at 0.3 . Then, we'll use Taylor's theorem to estimate the error.

```
Clear[f, x, T3, T5, T7, T9]
T3[x_] = Series[Sin[x], {x, 0, 3}] // Normal
T3[0.3]
```

Let's use Taylor's theorem to find the remainder. So far, we know $f(x) = \sin(x)$, $x = .3$, $n = 3$, so let's plug that into our equation for R_n.

$$\begin{aligned} R_n(x) &= \frac{f^{(n+1)}(c)}{(n+1)!}(x-a)^{n+1} \\ R_3(x) &= \frac{f^{(3+1)}(c)}{(3+1)!}(x-0)^{3+1} \\ &= \frac{\sin(c)}{4!}(.3)^4 \end{aligned}$$

for some c between 0 and 0.3 . Let's plug this into Mathematica:

```
Sin[c]/(4!)*(.3^4)
```

which should give us `0.0003375 Sin[c]`. The only thing we know about c is that c is somewhere between 0 and 0.3, so we can use this information to find an upper bound.

Since $\sin(x)$ is increasing on the interval $0 \le x \le 0.3$, and since $\sin(x) \le x$ on this interval, we can say

$$\begin{aligned}|R_n(x)| = 0.0003375\sin(c) \le 0.0003375\sin(0.3) < 0.0003375(0.3) \\ = 0.00010125\end{aligned}$$

To check this, let's use Mathematica to calculate the difference between $\sin(0.3)$ and the third degree Taylor approximation

```
Abs[Sin[0.3]-T3[0.3]]
```

Mathematica tells us the difference is `0.0000202067`, much smaller than our upper bound estimate.

Uniform Error Estimate

Now, suppose we want to approximate a function $f(x)$ over an interval $[a-r, a+r]$. As a consequence of Taylor's theorem, we have the uniform error estimate:

$$|R_n(x)| \le \frac{Mr^{n+1}}{(n+1)!} \qquad \text{for} \quad a-r \le x \le a+r$$

where M is the maximum value of $\left|f^{(n+1)}(x)\right|$ on the interval $[a-r, a+r]$.

Uniform Error Estimate Example

Let's try to find a Taylor polynomial approximation of $\cos(x)$ on the interval $[0, \pi/2]$ with an error less than 0.00005. Note that $\cos(x)$ and all of its derivatives have values between -1 and 1, so we can set $M = 1$. Plugging back into our error estimate inequality:

$$|R_n(x)| \le \frac{(1)(\pi/4)^{n+1}}{(n+1)!}$$

So we want to find the smallest value of n that satisfies

$$\frac{(\pi/4)^{n+1}}{(n+1)!} \le 0.00005$$

We'll use Mathematica to create a table of values:

```
Table[{n, ((Pi/4.)^(n+1))/((n+1)!)}, {n, 4, 8}] // TableForm // N
```

Your output should look like this:

```
4 0.00249039
5 0.000325999
6 0.0000365762
7 3.59086 x 10^-6
8 3.13362 x 10^-7
```

The first value of n where our error estimate is less than 0.00005 is $n = 6$. So, we can create a function for the sixth-degree Taylor polynomial and check how it compares to $\cos(x)$ on a plot.

```
T6[x_] = Series[Cos[x], {x, Pi/4, 6}] // Normal
Plot[{T6[x], Cos[x]}, {x, 0, Pi/2}]
Plot[{T6[x], Cos[x]}, {x, -5, 6}]
```

If you'll notice, the functions are virtually identical inside the region $[0, \pi/2]$, which is what we wanted to get.

Question 1

Find the fifth- and tenth-degree Taylor polynomials for

$$f(x) = \cos(1 - e^x)$$

about $x = 1$ and plot them along with $\cos(1 - e^x)$ on the interval $[-2, 3]$. (Note: The plot looks nicer if you set `PlotRange` to {`-2, 2`}. Also, it helps to put in `PlotLegends -> "Expressions"`)

Question 2

Find the third-degree Taylor polynomials for

$$f(x) = (1 - 3x)^{-5}$$

about two centers: $x = 1/6$ and $x = 1/2$.
Plot them along with $f(x)$ on the same interval $[0.15, 0.5]$. Now for the series with $x = 1/6$ as its center, use Mathematica function `SeriesCoefficient` to generate a table of Maclaurin coefficients. Use `FindSequenceFunction` to find a general formula for k-th term. Then determine the series' convergence interval using `SumConvergence`. (Warning: SumConvergence only works with the general term formula, not the sum itself)

Question 3

Find a polynomial approximation to

$$f(x) = e^x$$

centered at $x = 0$ on the interval [-2, 2]. Use the Uniform Error Estimate criteria to determine which value of n gives an approximation with an error less than 0.005 on this interval. Plot both the polynomial approximation and e^x on [-10, 10].